Applications of the J-integral to dynamical problems in geotechnical engineering

This notebook contains details of and verifications on the derivation and results displayed in the article. References to the actual body of the text are included. Here we use *z* where *z/H* appears in the text:

Solution for the far-field displacement field

In order to evaluate the right-hand sides of eq.(21) and the rest of expressions in that involve J_∞, it is necessary to quantify the displacement field in the far-field across the stratum.

The total displacement is normalized with the imposed base displacement, X_g, the distances with the wall height *H*.

All the equation are already written in dimensionless fashion.
The dimensionless frequency $r = \omega \mathrm{H}/\sqrt{(\mu/\rho)}$, the dimensionless distance measured from the top is $\xi = 1 - z/H$.

Displacement field in the far-field:

This corresponds to eq.(22):

```
In[ ]:= U[ξ_, r_] = Cos[r * ξ] / Cos[r];
```

$$U[\xi\_, r\_] = \frac{\mathrm{Cos}[r * \xi]}{\mathrm{Cos}[r]};$$

We have to obtain the value of the integral (21), which is divided in two parts:

Contribution of elastic gradients

In[ ]:=
```
Refine[(1/(r*(1 - i*δ/2)))^2 *
  Integrate[Abs[D[U[x, r*(1 - i*δ/2)], x] /. x -> ξ]^2, {ξ, 0, 1}], {r > 0, δ > 0}]
```

Out[ ]=
$$\frac{(4+\delta^2)\,(-r\,\delta\,\mathrm{Sin}[2\,r]+2\,r\,\mathrm{Sinh}[r\,\delta])}{8\,r^2\,\left(1-\frac{i\,\delta}{2}\right)^2\,\delta\,(\mathrm{Cos}[2\,r]+\mathrm{Cosh}[r\,\delta])}$$

See that the damping is already introduced. Likewise:

Contribution of kinetic (total) energy

We specify “total” kinetic energy as later we will have to subtract the contribution of rigid body movement.

In[ ]:=
```
Simplify@Refine[Integrate[Abs[U[ξ, r*(1 - i*δ/2)]]^2, {ξ, 0, 1}], {r > 0, δ > 0}]
```

Out[ ]=
$$\frac{\delta\,\mathrm{Cos}[r]\,\mathrm{Sin}[r]+\mathrm{Sinh}[r\,\delta]}{r\,\delta\,\mathrm{Cos}[2\,r]+r\,\delta\,\mathrm{Cosh}[r\,\delta]}$$

Combining and subtracting rigid body kinetic energy

In[ ]:=
$$\mathrm{FullSimplify}\left[-1+\frac{\delta\,\mathrm{Cos}[r]\,\mathrm{Sin}[r]+\mathrm{Sinh}[r\,\delta]}{r\,\delta\,\mathrm{Cos}[2\,r]+r\,\delta\,\mathrm{Cosh}[r\,\delta]}-\frac{(4+\delta^2)\,(-r\,\delta\,\mathrm{Sin}[2\,r]+2\,r\,\mathrm{Sinh}[r\,\delta])}{8\,r^2\,\left(1-\frac{i\,\delta}{2}\right)^2\,\delta\,(\mathrm{Cos}[2\,r]+\mathrm{Cosh}[r\,\delta])}\right]$$

Out[ ]=
$$\frac{-r\,(2\,i+\delta)\,\mathrm{Cos}[2\,r]-r\,(2\,i+\delta)\,\mathrm{Cosh}[r\,\delta]+2\,i\,\mathrm{Sin}[2\,r]+2\,\mathrm{Sinh}[r\,\delta]}{r\,(2\,i+\delta)\,(\mathrm{Cos}[2\,r]+\mathrm{Cosh}[r\,\delta])}$$

The latter still has to be divided by a $1/r^2$ factor coming from the non-dimensionalization process.

Approximate expression

This portion corresponds to the discussion in eq.(33), Section 3.3:

In[ ]:=
```
FullSimplify@Series[(-r (2 i + δ) Cos[2 r] - r (2 i + δ) Cosh[r δ] + 2 i Sin[2 r] + 2 Sinh[r δ]) /
   (r^3 (2 i + δ) (Cos[2 r] + Cosh[r δ])), {δ, 0, 2}]
```

Out[ ]=
$$\frac{-r+\mathrm{Tan}[r]}{r^3}+\frac{i\,\left(-r\,\mathrm{Sec}[r]^2+\mathrm{Tan}[r]\right)\,\delta}{2\,r^3}+\frac{\left(-\mathrm{Tan}[r]+r\,\mathrm{Sec}[r]^2\,(1-r\,\mathrm{Tan}[r])\right)\,\delta^2}{4\,r^3}+O[\delta]^3$$

Section 4.1: Earth Thrust Results

Auxiliary parameter: ratio between P and S wave velocities, i.e., the factor **c** that appears in many expressions:

In[ ]:=
```
c[ν_] = Sqrt[(2 * (1 - ν))/(1 - 2 * ν)];
```

Evaluation of the J-integral contribution at infinity, eq.(21) , via numerical integration:

In[ ]:=
```
BoundFEM[r_, δ_] := Abs[1/r^2 NIntegrate[-1 + Abs[U[ξ, r * (1 - ⅈ * δ / 2)]]^2 -
    (1/(r * (1 - ⅈ * δ / 2)))^2 * Abs[D[U[x, r * (1 - ⅈ * δ / 2)], x] /. x -> ξ]^2, {ξ, 0, 1}]];
```

The dynamic thrust bound, RHS of (32) evaluated numerically:

In[ ]:=
```
Qdynbound[δ_, ν_, r_] := Sqrt[(1 + c[ν]^2) BoundFEM[r, δ]];
```

4.1.1 Dynamic results for verification

List of dimensionless frequencies ($2\omega H / \pi c_s$) used to evaluate the results in the plots:

In[ ]:=
```
omList = Subdivide[0.1, 3.5, 250];
```

The following image displays some of the results contained in Figure 3 and Figure 4.
Set damping here to be DE=0.06:

In[ ]:=
```
DE = 0.06; ListLinePlot[{Qdynbound[DE, 0.1, π/2 #] & /@ omList,
   Qdynbound[DE, 0.33, π/2 #] & /@ omList,
   Qdynbound[DE, 0.4, π/2 #] & /@ omList},
  PlotRange → {Automatic, {0, 8}}, PlotLegends → {"ν=0.1", "ν=0.33", "ν=0.4"},
  AxesLabel → {Style[ω/ω_s, Large, Bold], Style[Q, Large, Bold]}]
```

Out[ ]=



Introduce also the approximation discussed in the text, eq.(39), and compare it to direct integration:

In[ ]:=
```
QdynboundApprox[ν_, r_] = (1 + c[ν]^2)^(1/2) * Abs@Sqrt[(r - Tan[r])/r^3];
```

In[ ]:=
```
DE = 0.06; ListLinePlot[{Qdynbound[DE, 0.4, π/2 #] & /@ omList,
   QdynboundApprox[0.4, π/2 #/Sqrt[1 + ⅈ * DE]] & /@ omList},
  PlotStyle → {Automatic, {Red, Dashed}},
  PlotRange → {Automatic, {0, 6}},
  PlotLegends → {"Numerical Integration Exact Bound", "Approximation"},
  AxesLabel → {Style[ω/ωs, Large, Bold], Style[Q, Large, Bold]}]
```

Out[ ]=



In[ ]:=
```
DE = 0.18; ListLinePlot[{Qdynbound[DE, 0.33, π/2 #] & /@ omList,
   QdynboundApprox[0.33, π/2 #/Sqrt[1 + ⅈ * DE]] & /@ omList},
  PlotStyle → {Automatic, {Red, Dashed}},
  PlotRange → {Automatic, {0, 6}},
  PlotLegends → {"Numerical Integration Exact Bound", "Approximation"},
  AxesLabel → {Style[ω/ωs, Large, Bold], Style[Q, Large, Bold]}]
```

Out[ ]=



4.1.2. Quasi-static results for verification

Long-wavelength limit:

In[ ]:=
```
Limit[QdynboundApprox[ν, r], r → 0]
```

Out[ ]=

$$\sqrt{\frac{3-4\nu}{3-6\nu}}$$

The quasi-static thrust bound in this case, eq.(35):

In[ ]:=
```
Qqsbound[ν_] = Sqrt[(1 + c[ν]^2)/3];
```

Display the bound outcome as in Figure 5 (influence of the compressibility):

In[ ]:=
```
Plot[Qqsbound[ν], {ν, 0, 0.45}, PlotRange → {Automatic, {0, 2}},
 PlotLegends → "Expressions", AxesLabel → {Style[ν, Large, Bold], Style[Q, Large, Bold]}]
```

Out[ ]=